
Computational Optimization for Economists

Sven Leyffer, Jorge Moré, and Todd Munson

Computational Optimization Overview

1. Introducion to Optimization [Moré]

2. Continuous Optimization in AMPL [Munson]

3. Optimization Software [Leyffer]

4. Complementarity & Games [Munson]

Part I

Introduction, Applications, and Formulations

Outline

• Software
• Views of optimization
• Characteristic of optimization software
• Case studies in optimization software

• Environments
• Modeling Languages: AMPL, GAMS
• Solving optimization problems
• Automatic differentiation

• Tools
• Benchmarking
• Performance profiles
• Scale invariance

Nonlinearly Constrained Optimization

min {f(x) : xl ≤ x ≤ xu, cl ≤ c(x) ≤ cu}

• Objective function is defined by f : Rn 7→ R
• Constraints are defined by c : Rn 7→ Rm.

• Bounds xl ≤ x ≤ xu on the variables x ∈ Rn.

• First-order algorithms require the gradient

∇f(x) = (∂if(x)), ∇c1(x), . . . ,∇cm(x)

• Second order algorithms require the Hessians

∇2f(x) = (∂i,jf(x)), ∇2c1(x), . . . ,∇2cm(x)

Part II

Continuous Optimization in AMPL

Modeling Languages

• Portable language for optimization problems
• Algebraic description
• Models easily modified and solved
• Large problems can be processed
• Programming language features

• Many available optimization algorithms
• No need to compile C/FORTRAN code
• Derivatives automatically calculated
• Algorithms specific options can be set

• Communication with other tools
• Relational databases and spreadsheets
• MATLAB interface for function evaluations

• Excellent documentation

• Large user communities

Model Declaration

• Sets
• Unordered, ordered, and circular sets
• Cross products and point to set mappings
• Set manipulation

• Parameters and variables
• Initial and default values
• Lower and upper bounds
• Check statements
• Defined variables

• Objective function and constraints
• Equality, inequality, and range constraints
• Complementarity constraints
• Multiple objectives

• Problem statement

Data and Commands

• Data declaration
• Set definitions

• Explicit list of elements
• Implicit list in parameter statements

• Parameter definitions
• Tables and transposed tables
• Higher dimensional parameters

• Execution commands
• Load model and data
• Select problem, algorithm, and options
• Solve the instance
• Output results

• Other operations
• Let and fix statements
• Conditionals and loop constructs
• Execution of external programs

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Model Formulation

• Economy with n agents and m commodities
• e ∈ <n×m are the endowments
• α ∈ <n×m and β ∈ <n×m are the utility parameters
• λ ∈ <n are the social weights

• Social planning problem

max
x≥0

n∑
i=1

λi

(
m∑

k=1

αi,k(1 + xi,k)
1−βi,k

1− βi,k

)

subject to
n∑

i=1

xi,k ≤
n∑

i=1

ei,k ∀k = 1, . . . ,m

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Model: social1.mod

param n > 0, integer; # Agents

param m > 0, integer; # Commodities

param e {1..n, 1..m} >= 0, default 1; # Endowment

param lambda {1..n} > 0; # Social weights

param alpha {1..n, 1..m} > 0; # Utility parameters

param beta {1..n, 1..m} > 0;

var x{1..n, 1..m} >= 0; # Consumption

var u{i in 1..n} = # Utility

sum {k in 1..m} alpha[i,k] * (1 + x[i,k])^(1 - beta[i,k]) / (1 - beta[i,k]);

maximize welfare:

sum {i in 1..n} lambda[i] * u[i];

subject to

consumption {k in 1..m}:

sum {i in 1..n} x[i,k] <= sum {i in 1..n} e[i,k];

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Data: social1.dat

param n := 3; # Agents

param m := 4; # Commodities

param alpha : 1 2 3 4 :=

1 1 1 1 1

2 1 2 3 4

3 2 1 1 5;

param beta (tr) : 1 2 3 :=

1 1.5 2 0.6

2 1.6 3 0.7

3 1.7 2 2.0

4 1.8 2 2.5;

param : lambda :=

1 1

2 1

3 1;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Commands: social1.cmd

# Load model and data

model social1.mod;

data social1.dat;

# Specify solver and options

option solver "minos";

option minos_options "outlev=1";

# Solve the instance

solve;

# Output results

display x;

printf {i in 1..n} "%2d: % 5.4e\n", i, u[i];

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Output

ampl: include social1.cmd;

MINOS 5.5: outlev=1

MINOS 5.5: optimal solution found.

25 iterations, objective 2.252422003

Nonlin evals: obj = 44, grad = 43.

x :=

1 1 0.0811471

1 2 0.574164

1 3 0.703454

1 4 0.267241

2 1 0.060263

2 2 0.604858

2 3 1.7239

2 4 1.47516

3 1 2.85859

3 2 1.82098

3 3 0.572645

3 4 1.2576

;

1: -5.2111e+00

2: -4.0488e+00

3: 1.1512e+01

ampl: quit;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Model: social2.mod

set AGENTS; # Agents

set COMMODITIES; # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param lambda {AGENTS} > 0; # Social weights

param alpha {AGENTS, COMMODITIES} > 0; # Utility parameters

param beta {AGENTS, COMMODITIES} > 0;

param gamma {i in AGENTS, k in COMMODITIES} := 1 - beta[i,k];

var x{AGENTS, COMMODITIES} >= 0; # Consumption

var u{i in AGENTS} = # Utility

sum {k in COMMODITIES} alpha[i,k] * (1 + x[i,k])^gamma[i,k] / gamma[i,k];

maximize welfare:

sum {i in AGENTS} lambda[i] * u[i];

subject to

consumption {k in COMMODITIES}:

sum {i in AGENTS} x[i,k] <= sum {i in AGENTS} e[i,k];

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Data: social2.dat

set COMMODITIES := Books, Cars, Food, Pens;

param: AGENTS : lambda :=

Jorge 1

Sven 1

Todd 1;

param alpha : Books Cars Food Pens :=

Jorge 1 1 1 1

Sven 1 2 3 4

Todd 2 1 1 5;

param beta (tr): Jorge Sven Todd :=

Books 1.5 2 0.6

Cars 1.6 3 0.7

Food 1.7 2 2.0

Pens 1.8 2 2.5;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Commands: social2.cmd

# Load model and data

model social2.mod;

data social2.dat;

# Specify solver and options

option solver "minos";

option minos_options "outlev=1";

# Solve the instance

solve;

# Output results

display x;

printf {i in AGENTS} "%5s: % 5.4e\n", i, u[i];

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Output

ampl: include social2.cmd

MINOS 5.5: outlev=1

MINOS 5.5: optimal solution found.

25 iterations, objective 2.252422003

Nonlin evals: obj = 44, grad = 43.

x :=

Jorge Books 0.0811471

Jorge Cars 0.574164

Jorge Food 0.703454

Jorge Pens 0.267241

Sven Books 0.060263

Sven Cars 0.604858

Sven Food 1.7239

Sven Pens 1.47516

Todd Books 2.85859

Todd Cars 1.82098

Todd Food 0.572645

Todd Pens 1.2576

;

Jorge: -5.2111e+00

Sven: -4.0488e+00

Todd: 1.1512e+01

ampl: quit;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Model Formulation

• Route commodities through a network
• N is the set of nodes
• A ⊆ N ×N is the set of arcs
• K is the set of commodities
• α and β are the congestion parameters
• b denotes the supply and demand

• Multicommodity network flow problem

max
x≥0,f≥0

∑
(i,j)∈A

(
αi,jfi,j + βi,jf

4
i,j

)
subject to

∑
(i,j)∈A

xi,j,k ≤
∑

(j,i)∈A

xj,i,k + bi,k ∀i ∈ N , k ∈ K

fi,j =
∑
k∈K

xi,j,k ∀(i, j) ∈ A

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Model: network.mod

set NODES; # Nodes in network

set ARCS within NODES cross NODES; # Arcs in network

set COMMODITIES := 1..3; # Commodities

param b {NODES, COMMODITIES} default 0; # Supply/demand

check {k in COMMODITIES}: # Supply exceeds demand

sum{i in NODES} b[i,k] >= 0;

param alpha{ARCS} >= 0; # Linear part

param beta{ARCS} >= 0; # Nonlinear part

var x{ARCS, COMMODITIES} >= 0; # Flow on arcs

var f{(i,j) in ARCS} = # Total flow

sum {k in COMMODITIES} x[i,j,k];

minimize time:

sum {(i,j) in ARCS} (alpha[i,j]*f[i,j] + beta[i,j]*f[i,j]^4);

subject to

conserve {i in NODES, k in COMMODITIES}:

sum {(i,j) in ARCS} x[i,j,k] <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Data: network.dat

set NODES := 1 2 3 4 5;

param: ARCS : alpha beta =

1 2 1 0.5

1 3 1 0.4

2 3 2 0.7

2 4 3 0.1

3 2 1 0.0

3 4 4 0.5

4 1 5 0.0

4 5 2 0.1

5 2 0 1.0;

let b[1,1] := 7; # Node 1, Commodity 1 supply

let b[4,1] := -7; # Node 4, Commodity 1 demand

let b[2,2] := 3; # Node 2, Commodity 2 supply

let b[5,2] := -3; # Node 5, Commodity 2 demand

let b[3,3] := 5; # Node 1, Commodity 3 supply

let b[1,3] := -5; # Node 4, Commodity 3 demand

fix {i in NODES, k in COMMODITIES: (i,i) in ARCS} x[i,i,k] := 0;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Commands: network.cmd

# Load model and data

model network.mod;

data network.dat;

# Specify solver and options

option solver "minos";

option minos_options "outlev=1";

# Solve the instance

solve;

# Output results

for {k in COMMODITIES} {

printf "Commodity: %d\n", k > network.out;

printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > network.out;

printf "\n" > network.out;

}

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Output

ampl: include network.cmd;

MINOS 5.5: outlev=1

MINOS 5.5: optimal solution found.

12 iterations, objective 1505.526478

Nonlin evals: obj = 14, grad = 13.

ampl: quit;

Leyffer, Moré, and Munson Computational Optimization



Overview
Examples

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Results: network.out

Commodity: 1

1.2 = 3.3775e+00

1.3 = 3.6225e+00

2.4 = 6.4649e+00

3.2 = 3.0874e+00

3.4 = 5.3510e-01

Commodity: 2

2.4 = 3.0000e+00

4.5 = 3.0000e+00

Commodity: 3

3.4 = 5.0000e+00

4.1 = 5.0000e+00

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Initial Coordinate Descent: wardrop0.cmd

# Load model and data

model network.mod;

data network.dat;

option solver "minos";

option minos_options "outlev=1";

# Coordinate descent method

fix {(i,j) in ARCS, k in COMMODITIES} x[i,j,k];

drop {i in NODES, k in COMMODITIES} conserve[i,k];

for {iter in 1..100} {

for {k in COMMODITIES} {

unfix {(i,j) in ARCS} x[i,j,k];

restore {i in NODES} conserve[i,k];

solve;

fix {(i,j) in ARCS} x[i,j,k];

drop {i in NODES} conserve[i,k];

}

}

# Output results

for {k in COMMODITIES} {

printf "\nCommodity: %d\n", k > network.out;

printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > network.out;

}

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Improved Coordinate Descent: wardrop.mod

set NODES; # Nodes in network

set ARCS within NODES cross NODES; # Arcs in network

set COMMODITIES := 1..3; # Commodities

param b {NODES, COMMODITIES} default 0; # Supply/demand

param alpha {ARCS} >= 0; # Linear part

param beta {ARCS} >= 0; # Nonlinear part

var x {ARCS, COMMODITIES} >= 0; # Flow on arcs

var f {(i,j) in ARCS} = # Total flow

sum {k in COMMODITIES} x[i,j,k];

minimize time {k in COMMODITIES}:

sum {(i,j) in ARCS} (alpha[i,j]*f[i,j] + beta[i,j]*f[i,j]^4);

subject to

conserve {i in NODES, k in COMMODITIES}:

sum {(i,j) in ARCS} x[i,j,k] <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];

problem subprob {k in COMMODITIES}: time[k], {i in NODES} conserve[i,k],

{(i,j) in ARCS} x[i,j,k], f;

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Improved Coordinate Descent: wardrop1.cmd

# Load model and data

model wardrop.mod;

data wardrop.dat;

# Specify solver and options

option solver "minos";

option minos_options "outlev=1";

# Coordinate descent method

for {iter in 1..100} {

for {k in COMMODITIES} {

solve subprob[k];

}

}

for {k in COMMODITIES} {

printf "Commodity: %d\n", k > wardrop.out;

printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > wardrop.out;

printf "\n" > wardrop.out;

}

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Final Coordinate Descent: wardrop2.cmd

# Load model and data

model wardrop.mod;

data wardrop.dat;

# Specify solver and options

option solver "minos";

option minos_options "outlev=1";

# Coordinate descent method

param xold{ARCS, COMMODITIES};

param xnew{ARCS, COMMODITIES};

repeat {

for {k in COMMODITIES} {

problem subprob[k];

let {(i,j) in ARCS} xold[i,j,k] := x[i,j,k];

solve;

let {(i,j) in ARCS} xnew[i,j,k] := x[i,j,k];

}

} until (sum {(i,j) in ARCS, k in COMMODITIES} abs(xold[i,j,k] - xnew[i,j,k]) <= 1e-6);

for {k in COMMODITIES} {

printf "Commodity: %d\n", k > wardrop.out;

printf {(i,j) in ARCS: x[i,j,k] > 0} "%d.%d = % 5.4e\n", i, j, x[i,j,k] > wardrop.out;

printf "\n" > wardrop.out;

}

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Ordered Sets

param V, integer; # Number of vertices

param E, integer; # Number of elements

set VERTICES := {1..V}; # Vertex indices

set ELEMENTS := {1..E}; # Element indices

set COORDS := {1..3} ordered; # Spatial coordinates

param T{ELEMENTS, 1..4} in VERTICES; # Tetrahedral elements

var x{VERTICES, COORDS}; # Position of vertices

var norm{e in ELEMENTS} = sum{i in COORDS, j in 1..4}

(x[T[e,j], i] - x[T[e,1], i])^2;

var area{e in ELEMENTS} = sum{i in COORDS}

(x[T[e,2], i] - x[T[e,1], i]) *

((x[T[e,3], nextw(i)] - x[T[e,1], nextw(i)]) *

(x[T[e,4], prevw(i)] - x[T[e,1], prevw(i)]) -

(x[T[e,3], prevw(i)] - x[T[e,1], prevw(i)]) *

(x[T[e,4], nextw(i)] - x[T[e,1], nextw(i)]));

minimize f: sum {e in ELEMENTS} norm[e] / max(area[e], 0) ^ (2 / 3);

Social Planning for Endowment Economy
Traffic Routing with Congestion
Finite Element Method

Circular Sets

param V, integer; # Number of vertices

param E, integer; # Number of elements

set VERTICES := {1..V}; # Vertex indices

set ELEMENTS := {1..E}; # Element indices

set COORDS := {1..3} circular; # Spatial coordinates

param T{ELEMENTS, 1..4} in VERTICES; # Tetrahedral elements

var x{VERTICES, COORDS}; # Position of vertices

var norm{e in ELEMENTS} = sum{i in COORDS, j in 1..4}

(x[T[e,j], i] - x[T[e,1], i])^2;

var area{e in ELEMENTS} = sum{i in COORDS}

(x[T[e,2], i] - x[T[e,1], i]) *

((x[T[e,3], next(i)] - x[T[e,1], next(i)]) *

(x[T[e,4], prev(i)] - x[T[e,1], prev(i)]) -

(x[T[e,3], prev(i)] - x[T[e,1], prev(i)]) *

(x[T[e,4], next(i)] - x[T[e,1], next(i)]));

minimize f: sum {e in ELEMENTS} norm[e] / max(area[e], 0) ^ (2 / 3);

Leyffer, Moré, and Munson Computational Optimization



Optimization Methods
Optimization Software

Beyond Nonlinear Optimization

Part III

Optimization Software

Optimization Software

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Generic Nonlinear Optimization Problem

Nonlinear Programming (NLP) problem
minimize

x
f(x) objective

subject to c(x) = 0 constraints
x ≥ 0 variables

• f : Rn → R, c : Rn → Rm smooth (typically C2)

• x ∈ Rn finite dimensional (may be large)

• more general l ≤ c(x) ≤ u possible

Optimization Software

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Optimality Conditions for NLP

Constraint qualification (CQ)
Linearizations of c(x) = 0 characterize all feasible perturbations
⇒ rules out cusps etc.

x∗ local minimizer & CQ holds ⇒ ∃ multipliers y∗, z∗:

∇f(x∗)−∇c(x∗)T y∗ − z∗ = 0
c(x∗) = 0
X∗z∗ = 0

x∗ ≥ 0, z∗ ≥ 0

where X∗ = diag(x∗), thus X∗z∗ = 0 ⇔ x∗i z
∗
i = 0

Lagrangian: L(x, y, z) := f(x)− yT c(x)− zT x

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Optimality Conditions for NLP

Objective gradient is linear combination of constraint gradients

g(x) = A(x)y, where g(x) := ∇f(x), A(x) := ∇c(x)T

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Newton’s Method for Nonlinear Equations

Solve F (x) = 0:
Get approx. xk+1 of solution of F (x) = 0
by solving linear model about xk:

F (xk) +∇F (xk)
T (x− xk) = 0

for k = 0, 1, . . .

Theorem: If F ∈ C2, and ∇F (x∗) nonsingular,
then Newton converges quadratically near x∗.

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Newton’s Method for Nonlinear Equations

Next: two classes of methods based on Newton ...

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Sequential Quadratic Programming (SQP)

Consider equality constrained NLP

minimize
x

f(x) subject to c(x) = 0

Optimality conditions:

∇f(x)−∇c(x)T y = 0 and
c(x) = 0

... system of nonlinear equations: F (w) = 0 for w = (x, y).

... solve using Newton’s method

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Sequential Quadratic Programming (SQP)

Nonlinear system of equations (KKT conditions)

∇f(x)−∇c(x)T y = 0 and c(x) = 0

Apply Newton’s method from wk = (xk, yk) ... Hk = ∇2L(xk, yk)[
Hk −Ak

AT
k 0

](
sx

sy

)
= −

(
∇xL(xk, yk)

ck

)
... set (xk+1, yk+1) = (xk + sx, yk + sy) ... Ak = ∇c(xk)

T

... solve for yk+1 = yk + sy directly instead:[
Hk −Ak

AT
k 0

](
s

yk+1

)
= −

(
∇fk

ck

)
... set (xk+1, yk+1) = (xk + s, yk+1)

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Sequential Quadratic Programming (SQP)

Newton’s Method for KKT conditions leads to:[
Hk −Ak

AT
k 0

](
s

yk+1

)
= −

(
∇fk

ck

)
... are optimality conditions of QP{

minimize
s

∇fT
k s + 1

2sT Hks

subject to ck + AT
k s = 0

... hence Sequential Quadratic Programming

Optimization Software

Beyond Nonlinear Optimization

Newton’s Method for Equations
Sequential Quadratic Programming
Interior Point Methods
Global Convergence

Sequential Quadratic Programming (SQP)

SQP for inequality constrained NLP:

minimize
x

f(x) subject to c(x) = 0 & x ≥ 0

REPEAT

1. Solve QP for (s, yk+1, zk+1)
minimize

s
∇fT

k s + 1
2sT Hks

subject to ck + AT
k s = 0

xk + s ≥ 0

2. Set xk+1 = xk + s

Modern Interior Point Methods (IPM)

General NLP

minimize
x

f(x) subject to c(x) = 0 & x ≥ 0

Perturbed µ > 0 optimality conditions (x, z > 0)

Fµ(x, y, z) =


∇f(x)−∇c(x)T y − z

c(x)
Xz − µe

 = 0

• Primal-dual formulation, where X = diag(x)

• Central path {x(µ), y(µ), z(µ) : µ > 0}
• Apply Newton’s method for sequence µ ↘ 0

Modern Interior Point Methods (IPM)

Newton’s method applied to primal-dual system ... ∇2Lk −Ak −I
AT

k 0 0
Zk 0 Xk

 ∆x
∆y
∆z

 = −Fµ(xk, yk, zk)

where Ak = ∇c(xk)
T , Xk diagonal matrix of xk.

Polynomial run-time guarantee for convex problems

Classical Interior Point Methods (IPM)

minimize
x

f(x) subject to c(x) = 0 & x ≥ 0

Related to classical barrier methods [Fiacco & McCormick]{
minimize

x
f(x)− µ

∑
log(xi)

subject to c(x) = 0

µ = 0.1 µ = 0.001

minimize x2
1 + x2

2 subject to x1 + x2
2 ≥ 1

Classical Interior Point Methods (IPM)

minimize
x

f(x) subject to c(x) = 0 & x ≥ 0

Relationship to barrier methods{
minimize

x
f(x)− µ

∑
log(xi)

subject to c(x) = 0

First order conditions

∇f(x)− µX−1e−A(x)y = 0
c(x) = 0

... apply Newton’s method ...

Classical Interior Point Methods (IPM)

Newton’s method for barrier problem from xk ...[
∇2Lk + µX−2

k −Ak

AT
k 0

](
∆x
∆y

)
= ...

Introduce Z(xk) := µX−1
k ... or ... Z(xk)Xk = µe[

∇2Lk + Z(xk)X
−1
k −Ak

Ak 0

](
∆x
∆y

)
= ...

... compare to primal-dual system ...

Classical Interior Point Methods (IPM)

Recall: Newton’s method applied to primal-dual system ... ∇2Lk −Ak −I
AT

k 0 0
Zk 0 Xk

 ∆x
∆y
∆z

 = −Fµ(xk, yk, zk)

Eliminate ∆z = −X−1Z∆x− Ze− µX−1e[
∇2Lk + ZkX

−1
k −Ak

Ak 0

](
∆x
∆y

)
= ...

Interior Point Methods (IPM)

Primal-dual system ...[
∇2Lk + ZkX

−1
k −Ak

Ak 0

](
∆x
∆y

)
= ...

... compare to barrier system ...[
∇2Lk + Z(xk)X

−1
k −Ak

Ak 0

](
∆x
∆y

)
= ...

• Zk is free, not Z(xk) = µX−1
k (primal multiplier)

• avoid difficulties with barrier ill-conditioning

Convergence from Remote Starting Points

minimize
x

f(x) subject to c(x) = 0 & x ≥ 0

• Newton’s method converges quadratically near a solution

• Newton’s method may diverge if started far from solution

• How can we safeguard against this failure?

... motivates penalty or merit functions that

1. monitor progress towards a solution

2. combine objective f(x) and constraint violation ‖c(x)‖

Penalty Functions (i)

Augmented Lagrangian Methods

minimize
x

L(x, yk, ρk) = f(x) − yT
k c(x) + 1

2ρk‖c(x)‖2

As yk → y∗: • xk → x∗ for ρk > ρ̄
• No ill-conditioning, improves convergence rate

• update ρk based on reduction in ‖c(x)‖2

• approx. minimize L(x, yk, ρk)

• first-order multiplier update: yk+1 = yk − ρkc(xk)
⇒ dual iteration

Penalty Functions (ii)

Exact Penalty Function

minimize
x

Φ(x, π) = f(x) + π‖c(x)‖

• combine constraints ad objective

• equivalence of optimality ⇒ exact for π > ‖y∗‖D

... now apply unconstrained techniques

• Φ nonsmooth, but equivalent to smooth problem (exercise)

... how do we enforce descent in merit functions???

Line Search Methods

SQP/IPM compute s descend direction: sT∇Φ < 0

Backtracking-Armijo line search

Given α0 = 1, β = 0.1, set l = 0

REPEAT

1. αl+1 = αl/2 & evaluate Φ(x + αl+1s)

2. l = l + 1

UNTIL Φ(x + αls) ≤ f(x) + αlβsT∇Φ

Converges to stationary point, or unbounded, or zero descend

Line Search Methods

Trust Region Methods

Globalize SQP (IPM) by adding trust region, ∆k > 0{
minimize

s
∇fT

k s + 1
2sT Hks

subject to ck + AT
k s = 0, xk + s ≥ 0, ‖s‖ ≤ ∆k

Trust Region Methods

Globalize SQP (IPM) by adding trust region, ∆k > 0{
minimize

s
∇fT

k s + 1
2sT Hks

subject to ck + AT
k s = 0, xk + s ≥ 0, ‖s‖ ≤ ∆k

REPEAT

1. Solve QP approximation about xk

2. Compute actual/predicted reduction, rk

3. IF rk ≥ 0.75 THEN xk+1 = xk + s increase ∆
ELSEIF rk ≥ 0.25 THEN xk+1 = xk + s
ELSE xk+1 = xk & decrease ∆ reject step

UNTIL convergence

Filter Methods for NLP

Penalty function can be inefficient

• Penalty parameter not known a priori

• Large penalty parameter ⇒ slow convergence

Filter Methods for NLP

Penalty function can be inefficient

• Penalty parameter not known a priori

• Large penalty parameter ⇒ slow convergence

Two competing aims in optimization:

1. Minimize f(x)

2. Minimize h(x) := ‖c(x)‖ ... more important

⇒ concept from multi-objective optimization:
(hk+1, fk+1) dominates (hl, fl) iff hk+1 ≤ hl & fk+1 ≤ fl

Filter Methods for NLP

Filter F : list of non-dominated pairs (hl, fl)

• new xk+1 acceptable to filter F , iff

1. hk+1 ≤ hl ∀l ∈ F , or
2. fk+1 ≤ fl ∀l ∈ F

• remove redundant entries

• reject new xk+1,
if hk+1 > hl & fk+1 > fl

... reduce trust region radius ∆ = ∆/2

⇒ often accept new xk+1, even if penalty function increases

Sequential Quadratic Programming

• filterSQP
• trust-region SQP; robust QP solver
• filter to promote global convergence

• SNOPT
• line-search SQP; null-space CG option
• `1 exact penalty function

• SLIQUE (part of KNITRO)
• SLP-EQP (“SQP” for larger problems)
• trust-region with `1 penalty

Other Methods: CONOPT generalized reduced gradient method

Interior Point Methods

• IPOPT (free: part of COIN-OR)
• line-search filter algorithm
• 2nd order convergence analysis for filter

• KNITRO
• trust-region Newton to solve barrier problem
• `1 penalty barrier function
• Newton system: direct solves or null-space CG

• LOQO
• line-search method
• Cholesky factorization; no convergence analysis

Other solvers: MOSEK (unsuitable or nonconvex problem)

Augmented Lagrangian Methods

• LANCELOT
• minimize augmented Lagrangian subject to bounds
• trust-region to force convergence
• iterative (CG) solves

• MINOS
• minimize augmented Lagrangian subject to linear constraints
• line-search; recent convergence analysis
• direct factorization of linear constraints

• PENNON
• suitable for semi-definite optimization
• alternative penalty terms

Automatic Differentiation

How do I get the derivatives ∇c(x), ∇2c(x) etc?

• hand-coded derivatives are error prone

• finite differences ∂ci(x)
∂xj

' ci(x+δej)−ci(x)
δ can be dangerous

where ej = (0, . . . , 0, 1, 0, . . . , 0) is jth unit vector

Automatic Differentiation

• chain rule techniques to differentiate program

• recursive application ⇒ “exact” derivatives

• suitable for huge problems, see www.autodiff.org

... already done for you in AMPL/GAMS etc.

Something Under the Bed is Drooling

1. floating point (IEEE) exceptions

2. unbounded problems

2.1 unbounded objective
2.2 unbounded multipliers

3. (locally) inconsistent problems

4. suboptimal solutions

... identify problem & suggest remedies

Floating Point (IEEE) Exceptions

Bad example: minimize barrier function

param mu default 1;
var x{1..2} >= -10, <= 10;
var s;
minimize barrier: x[1]^2 + x[2]^2 - mu*log(s);
subject to

cons: s = x[1] + x[2]^2 - 1;

... results in error message like
Cannot evaluate objective at start

... change initialization of s:
var s := 1; ... difficult, if IEEE during solve ...

Unbounded Objective

Penalized MPEC π = 1:

minimize
x

x2
1 + x2

2 − 4x1x2 + πx1x2

subject to x1, x2 ≥ 0

... unbounded below for all π < 2

Locally Inconsistent Problems

NLP may have no feasible point

feasible set: intersection of circles

Locally Inconsistent Problems

NLP may have no feasible point

var x{1..2} >= -1;
minimize objf: -1000*x[2];
subject to

con1: (x[1]+2)^2 + x[2]^2 <= 1;
con2: (x[1]-2)^2 + x[2]^2 <= 1;

• not all solvers recognize this ...

• finding feasible point ⇔ global optimization

Locally Inconsistent Problems

LOQO

| Primal | Dual
Iter | Obj Value Infeas | Obj Value Infeas
- - - - - - - - - - - - - - - - - - - - - - - - - - - - -
1 -1.000000e+03 4.2e+00 -6.000000e+00 1.0e-00

[...]
500 2.312535e-04 7.9e-01 1.715213e+12 1.5e-01
LOQO 6.06: iteration limit

... fails to converge ... not useful for user

dual unbounded →∞ ⇒ primal infeasible

Locally Inconsistent Problems

FILTER

iter | rho | ||d|| | f / hJ | ||c||/hJt

------+----------+------------+------------+------------

0:0 10.0000 0.00000 -1000.0000 16.000000

1:1 10.0000 2.00000 -1000.0000 8.0000000

[...]

8:2 2.00000 0.320001E-02 7.9999693 0.10240052E-04

9:2 2.00000 0.512000E-05 8.0000000 0.26214586E-10

filterSQP: Nonlinear constraints locally infeasible

... fast convergence to minimum infeasibility

... identify “blocking” constraints ... modify model/data

Locally Inconsistent Problems

Remedies for locally infeasible problems:

1. check your model: print constraints & residuals, e.g.
solve;
display conname, con.lb, con.body, con.ub;
display varname, var.lb, var, var.ub;
... look at violated and active constraints

2. try different nonlinear solvers (easy with AMPL)

3. build-up model from few constraints at a time

4. try different starting points ... global optimization

Suboptimal Solution & Multi-start

Problems can have many local mimimizers

param pi := 3.1416;
param n integer, >= 0, default 2;
set N := 1..n;
var x{N} >= 0, <= 32*pi, := 1;
minimize objf:
- sum{i in N} x[i]*sin(sqrt(x[i]));

default start point converges to local minimizer

Suboptimal Solution & Multi-start

param nD := 5; # discretization
set D := 1..nD;
param hD := 32*pi/(nD-1);
param optval{D,D};
model schwefel.mod; # load model

for {i in D}{
let x[1] := (i-1)*hD;
for {j in D}{

let x[2] := (j-1)*hD;
solve;
let optval[i,j] := objf;

}; # end for
}; # end for

Suboptimal Solution & Multi-start

display optval;
optval [*,*]
: 1 2 3 4 5 :=
1 0 24.003 -36.29 -50.927 56.909
2 24.003 -7.8906 -67.580 -67.580 -67.580
3 -36.29 -67.5803 -127.27 -127.27 -127.27
4 -50.927 -67.5803 -127.27 -127.27 -127.27
5 56.909 -67.5803 -127.27 -127.27 -127.27
;

... there exist better multi-start procedures

Optimization with Integer Variables

• modeling discrete choices ⇒ 0− 1 variables

• modeling integer decisions ⇒ integer variables
e.g. number of different stocks in portfolio (8-10)
not number of beers sold at Goose Island (millions)

⇒ Mixed Integer Nonlinear Program (MINLP)
MINLP solvers:

• branch (separate zi = 0 and zi = 1) and cut

• solve millions of NLP relaxations: MINLPBB, SBB

• outer approximation: iterate MILP and NLP solvers
BONMIN soon on COIN-OR

Portfolio Management

• N : Universe of asset to purchase

• xi: Amount of asset i to hold

• B: Budget

min
x∈R|N|

+

{
u(x) |

∑
i∈N

xi = B

}

• Markowitz: u(x)
def
= −αT x + λxT Qx

• α: Expected returns
• Q: Variance-covariance matrix of expected returns
• λ: Risk aversion parameter

More Realistic Models

• b ∈ R|N | of “benchmark” holdings

• Benchmark Tracking: u(x)
def
= (x− b)T Q(x− b)

• Constraint on E[Return]: αT x ≥ r

• Limit Names: |i ∈ N : xi > 0| ≤ K
• Use binary indicator variables to model the implication

xi > 0 ⇒ yi = 1
• Implication modeled with variable upper bounds:

xi ≤ Byi ∀i ∈ N

•
∑

i∈N yi ≤ K

Even More Models

• Min Holdings: (xi = 0) ∨ (xi ≥ m)
• Model implication: xi > 0 ⇒ xi ≥ m
• xi > 0 ⇒ yi = 1 ⇒ xi ≥ m
• xi ≤ Byi, xi ≥ myi ∀i ∈ N

• Round Lots: xi ∈ {kLi, k = 1, 2, . . .}
• xi − ziLi = 0, zi ∈ Z+ ∀i ∈ N

• Vector h of initial holdings

• Transactions: ti = |xi − hi|
• Turnover:

∑
i∈N ti ≤ ∆

• Transaction Costs:
∑

i∈N citi in objective

• Market Impact:
∑

i∈N γit
2
i in objective

Global Optimization

I need to find the GLOBAL minimum!

• use any NLP solver (often work well!)

• use the multi-start trick from previous slides

• global optimization based on branch-and-reduce: BARON
• constructs global underestimators
• refines region by branching
• tightens bounds by solving LPs
• solve problems with 100s of variables

• “voodoo” solvers: genetic algorithm & simulated annealing
no convergence theory ... usually worse than deterministic

Derivative-Free Optimization

My model does not have derivatives!

• Change your model ... good models have derivatives!

• pattern-search methods for min f(x)
• evaluate f(x) at stencil xk + ∆M
• move to new best point
• extend to NLP; some convergence theory
• matlab: NOMADm.m; parallel APPSPACK

• solvers based on building quadratic models

• “voodoo” solvers: genetic algorithm & simulated annealing
no convergence theory ... usually worse than deterministic

COIN-OR

http://www.coin-or.org

• COmputational INfrastructure for Operations Research

• A library of (interoperable) software tools for optimization

• A development platform for open source projects in the OR
community

• Possibly Relevant Modules:
• OSI: Open Solver Interface
• CGL: Cut Generation Library
• CLP: Coin Linear Programming Toolkit
• CBC: Coin Branch and Cut
• IPOPT: Interior Point OPTimizer for NLP
• NLPAPI: NonLinear Programming API

Definition
• Non-cooperative game played by n individuals

• Each player selects a strategy to optimize their objective
• Strategies for the other players are fixed

• Equilibrium reached when no improvement is possible
• Characterization of two player equilibrium (x∗, y∗)

x∗ ∈

{
arg min

x≥0
f1(x, y∗)

subject to c1(x) ≤ 0

y∗ ∈

{
arg min

y≥0
f2(x

∗, y)

subject to c2(y) ≤ 0

• Many applications in economics
• Bi-matrix games
• Cournot duopoly models
• General equilibrium models
• Arrow-Debreau models

Complementarity Formulation

• Assume each optimization problem is convex
• f1(·, y) is convex for each y
• f2(x, ·) is convex for each x
• c1(·) and c2(·) satisfy constraint qualification

• Then the first-order conditions are necessary and sufficient
min
x≥0

f1(x, y∗)

subject to c1(x) ≤ 0
⇔ 0 ≤ x ⊥ ∇xf1(x, y∗) + λT

1 ∇xc1(x) ≥ 0
0 ≤ λ1 ⊥ −c1(x) ≥ 0

min
y≥0

f2(x
∗, y)

subject to c2(y) ≤ 0
⇔ 0 ≤ y ⊥ ∇yf2(x

∗, y) + λT
2 ∇yc2(y) ≥ 0

0 ≤ λ2 ⊥ −c2(y) ≥ 0

0 ≤ x ⊥ ∇xf1(x, y) + λT
1 ∇xc1(x) ≥ 0

0 ≤ y ⊥ ∇yf2(x, y) + λT
2 ∇yc2(y) ≥ 0

0 ≤ λ1 ⊥ −c1(y) ≥ 0
0 ≤ λ2 ⊥ −c2(y) ≥ 0

• Nonlinear complementarity problem

• Square system – number of variables and constraints the same
• Each solution is an equilibrium for the Nash game

Formulation
• Firm f ∈ F chooses output xf to maximize profit

• u is the utility function

u =

1 +
∑
f∈F

xα
f


η
α

• α and η are parameters
• cf is the unit cost for each firm

• In particular, for each firm f ∈ F

x∗f ∈ arg max
xf≥0

(
∂u

∂xf
− cf

)
xf

• First-order optimality conditions

0 ≤ xf ⊥ cf − ∂u
∂xf

− xf
∂2u
∂x2

f
≥ 0

Model: oligopoly.mod

set FIRMS; # Firms in problem

param c {FIRMS}; # Unit cost

param alpha > 0; # Constants

param eta > 0;

var x {FIRMS} default 0.1; # Output (no bounds!)

var s = 1 + sum {f in FIRMS} x[f]^alpha; # Summation term

var u = s^(eta/alpha); # Utility

var du {f in FIRMS} = # Marginal price

eta * s^(eta/alpha - 1) * x[f]^(alpha - 1);

var dudu {f in FIRMS} = # Derivative

eta * (eta - alpha) * s^(eta/alpha - 2) * x[f]^(2 * alpha - 2) +

eta * (alpha - 1 ) * s^(eta/alpha - 1) * x[f]^( alpha - 2);

compl {f in FIRMS}:

0 <= x[f] complements c[f] - du[f] - x[f] * dudu[f] >= 0;

Data: oligopoly.dat

param: FIRMS : c :=
1 0.07
2 0.08
3 0.09;

param alpha := 0.999;
param eta := 0.2;

Commands: oligopoly.cmd

# Load model and data

model oligopoly.mod;

data oligopoly.dat;

# Specify solver and options

option presolve 0;

option solver "pathampl";

# Solve complementarity problem

solve;

# Output the results

printf {f in FIRMS} "Output for firm %2d: % 5.4e\n", f, x[f] > oligcomp.out;

Results: oligopoly.out

Output for firm 1: 8.3735e-01
Output for firm 2: 5.0720e-01
Output for firm 3: 1.7921e-01

Model Formulation
• Economy with n agents and m commodities

• e ∈ <n×m are the endowments
• α ∈ <n×m and β ∈ <n×m are the utility parameters
• p ∈ <m are the commodity prices

• Agent i maximizes utility with budget constraint

max
xi,∗≥0

m∑
k=1

αi,k(1 + xi,k)
1−βi,k

1− βi,k

subject to
m∑

k=1

pk (xi,k − ei,k) ≤ 0

• Market k sets price for the commodity

0 ≤ pk ⊥
n∑

i=1

(ei,k − xi,k) ≥ 0

Model: cge.mod

set AGENTS; # Agents

set COMMODITIES; # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param alpha {AGENTS, COMMODITIES} > 0; # Utility parameters

param beta {AGENTS, COMMODITIES} > 0;

var x {AGENTS, COMMODITIES}; # Consumption (no bounds!)

var l {AGENTS}; # Multipliers (no bounds!)

var p {COMMODITIES}; # Prices (no bounds!)

var du {i in AGENTS, k in COMMODITIES} = # Marginal prices

alpha[i,k] / (1 + x[i,k])^beta[i,k];

subject to

optimality {i in AGENTS, k in COMMODITIES}:

0 <= x[i,k] complements -du[i,k] + p[k] * l[i] >= 0;

budget {i in AGENTS}:

0 <= l[i] complements sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

market {k in COMMODITIES}:

0 <= p[k] complements sum {i in AGENTS} (e[i,k] - x[i,k]) >= 0;

Data: cge.dat

set AGENTS := Jorge, Sven, Todd;

set COMMODITIES := Books, Cars, Food, Pens;

param alpha : Books Cars Food Pens :=

Jorge 1 1 1 1

Sven 1 2 3 4

Todd 2 1 1 5;

param beta (tr): Jorge Sven Todd :=

Books 1.5 2 0.6

Cars 1.6 3 0.7

Food 1.7 2 2.0

Pens 1.8 2 2.5;

Commands: cge.cmd

# Load model and data

model cge.mod;

data cge.dat;

# Specify solver and options

option presolve 0;

option solver "pathampl";

# Solve the instance

solve;

# Output results

printf {i in AGENTS, k in COMMODITIES} "%5s %5s: % 5.4e\n", i, k, x[i,k] > cge.out;

printf "\n" > cge.out;

printf {k in COMMODITIES} "%5s: % 5.4e\n", k, p[k] > cge.out;

Results: cge.out

Jorge Books: 8.9825e-01

Jorge Cars: 1.4651e+00

Jorge Food: 1.2021e+00

Jorge Pens: 6.8392e-01

Sven Books: 2.5392e-01

Sven Cars: 7.2054e-01

Sven Food: 1.6271e+00

Sven Pens: 1.4787e+00

Todd Books: 1.8478e+00

Todd Cars: 8.1431e-01

Todd Food: 1.7081e-01

Todd Pens: 8.3738e-01

Books: 1.0825e+01

Cars: 6.6835e+00

Food: 7.3983e+00

Pens: 1.1081e+01

Commands: cgenum.cmd

# Load model and data

model cge.mod;

data cge.dat;

# Specify solver and options

option presolve 0;

option solver "pathampl";

# Solve the instance

drop market[’Books’];

fix p[’Books’] := 1;

solve;

# Output results

printf {i in AGENTS, k in COMMODITIES} "%5s %5s: % 5.4e\n", i, k, x[i,k] > cgenum.out;

printf "\n" > cgenum.out;

printf {k in COMMODITIES} "%5s: % 5.4e\n", k, p[k] > cgenum.out;

Results: cgenum.out

Jorge Books: 8.9825e-01

Jorge Cars: 1.4651e+00

Jorge Food: 1.2021e+00

Jorge Pens: 6.8392e-01

Sven Books: 2.5392e-01

Sven Cars: 7.2054e-01

Sven Food: 1.6271e+00

Sven Pens: 1.4787e+00

Todd Books: 1.8478e+00

Todd Cars: 8.1431e-01

Todd Food: 1.7081e-01

Todd Pens: 8.3738e-01

Books: 1.0000e+00

Cars: 6.1742e-01

Food: 6.8345e-01

Pens: 1.0237e+00

Formulation
• Players select strategies to minimize loss

• p ∈ <n is the probability player 1 chooses each strategy
• q ∈ <m is the probability player 2 chooses each strategy
• A ∈ <n×m is the loss matrix for player 1
• B ∈ <n×m is the loss matrix for player 2

• Optimization problem for player 1

min
0≤p≤1

pT Aq

subject to eT p = 1

• Optimization problem for player 2

min
0≤q≤1

pT Bq

subject to eT q = 1

• Complementarity problem

0 ≤ p ≤ 1 ⊥ Aq − λ1

0 ≤ q ≤ 1 ⊥ BT p− λ2

λ1 free ⊥ eT p = 1
λ2 free ⊥ eT q = 1

Model: bimatrix1.mod

param n > 0, integer; # Strategies for player 1

param m > 0, integer; # Strategies for player 2

param A{1..n, 1..m}; # Loss matrix for player 1

param B{1..n, 1..m}; # Loss matrix for player 2

var p{1..n}; # Probability player 1 selects strategy i

var q{1..m}; # Probability player 2 selects strategy j

var lambda1; # Multiplier for constraint

var lambda2; # Multiplier for constraint

subject to

opt1 {i in 1..n}: # Optimality conditions for player 1

0 <= p[i] <= 1 complements sum{j in 1..m} A[i,j] * q[j] - lambda1;

opt2 {j in 1..m}: # Optimality conditions for player 2

0 <= q[j] <= 1 complements sum{i in 1..n} B[i,j] * p[i] - lambda2;

con1:

lambda1 complements sum{i in 1..n} p[i] = 1;

con2:

lambda2 complements sum{j in 1..m} q[j] = 1;

Model: bimatrix2.mod

param n > 0, integer; # Strategies for player 1

param m > 0, integer; # Strategies for player 2

param A{1..n, 1..m}; # Loss matrix for player 1

param B{1..n, 1..m}; # Loss matrix for player 2

var p{1..n}; # Probability player 1 selects strategy i

var q{1..m}; # Probability player 2 selects strategy j

var lambda1; # Multiplier for constraint

var lambda2; # Multiplier for constraint

subject to

opt1 {i in 1..n}: # Optimality conditions for player 1

0 <= p[i] complements sum{j in 1..m} A[i,j] * q[j] - lambda1 >= 0;

opt2 {j in 1..m}: # Optimality conditions for player 2

0 <= q[j] complements sum{i in 1..n} B[i,j] * p[i] - lambda2 >= 0;

con1:

0 <= lambda1 complements sum{i in 1..n} p[i] >= 1;

con2:

0 <= lambda2 complements sum{j in 1..m} q[j] >= 1;

Leyffer, Moré, and Munson Computational Optimization



Nash Games
Stackelberg Games

Introduction
Oligopoly Model
Equilibrium for Endowment Economy
Bimatrix Games

Model: bimatrix3.mod

param n > 0, integer; # Strategies for player 1

param m > 0, integer; # Strategies for player 2

param A{1..n, 1..m}; # Loss matrix for player 1

param B{1..n, 1..m}; # Loss matrix for player 2

var p{1..n}; # Probability player 1 selects strategy i

var q{1..m}; # Probability player 2 selects strategy j

subject to

opt1 {i in 1..n}: # Optimality conditions for player 1

0 <= p[i] complements sum{j in 1..m} A[i,j] * q[j] >= 1;

opt2 {j in 1..m}: # Optimality conditions for player 2

0 <= q[j] complements sum{i in 1..n} B[i,j] * p[i] >= 1;

Pitfalls

• Nonsquare systems
• Side variables
• Side constraints

• Orientation of equations
• Skew symmetry preferred
• Proximal point perturbation

• AMPL presolve
• option presolve 0;

Definition
• Leader-follower game

• Dominant player (leader) selects a strategy y∗

• Then followers respond by playing a Nash game

x∗i ∈

{
arg min

xi≥0
fi(x, y)

subject to ci(xi) ≤ 0

• Leader solves optimization problem with equilibrium
constraints

min
y≥0,x,λ

g(x, y)

subject to h(y) ≤ 0
0 ≤ xi ⊥ ∇xifi(x, y) + λT

i ∇xici(xi) ≥ 0
0 ≤ λi ⊥ −ci(xi) ≥ 0

• Many applications in economics
• Optimal taxation
• Tolling problems

Nonlinear Programming Formulation

min
x,y,λ,s,t≥0

g(x, y)

subject to h(y) ≤ 0
si = ∇xifi(x, y) + λT

i ∇xici(xi)
ti = −ci(xi)∑

i

(
sT
i xi + λiti

)
≤ 0

• Constraint qualification fails
• Lagrange multiplier set unbounded
• Constraint gradients linearly dependent
• Central path does not exist

• Able to prove convergence results for some methods

• Reformulation very successful and versatile in practice

Penalization Approach

min
x,y,λ,s,t≥0

g(x, y) + π
∑

i

(
sT
i xi + λiti

)
subject to h(y) ≤ 0

si = ∇xifi(x, y) + λT
i ∇xici(xi)

ti = −ci(xi)

• Optimization problem satisfies constraint qualification

• Need to increase π

Relaxation Approach

min
x,y,λ,s,t≥0

g(x, y)

subject to h(y) ≤ 0
si = ∇xifi(x, y) + λT

i ∇xici(xi)
ti = −ci(xi)∑

i

(
sT
i xi + λiti

)
≤ τ

• Need to decrease τ

Model Formulation
• Economy with n agents and m commodities

• e ∈ <n×m are the endowments
• α ∈ <n×m and β ∈ <n×m are the utility parameters
• p ∈ <m are the commodity prices

• Agent i maximizes utility with budget constraint

max
xi,∗≥0

m∑
k=1

αi,k(1 + xi,k)
1−βi,k

1− βi,k

subject to
m∑

k=1

pk (xi,k − ei,k) ≤ 0

• Market k sets price for the commodity

0 ≤ pk ⊥
n∑

i=1

(ei,k − xi,k) ≥ 0

Model: cgempec.mod

set LEADER; # Leader

set FOLLOWERS; # Followers

set AGENTS := LEADER union FOLLOWERS; # All the agents

check: (card(LEADER) == 1 && card(LEADER inter FOLLOWERS) == 0);

set COMMODITIES; # Commodities

param e {AGENTS, COMMODITIES} >= 0, default 1; # Endowment

param alpha {AGENTS, COMMODITIES} > 0; # Utility parameters

param beta {AGENTS, COMMODITIES} > 0;

var x {AGENTS, COMMODITIES}; # Consumption (no bounds!)

var l {FOLLOWERS}; # Multipliers (no bounds!)

var p {COMMODITIES}; # Prices (no bounds!)

var u {i in AGENTS} = # Utility

sum {k in COMMODITIES} alpha[i,k] * (1 + x[i,k])^(1 - beta[i,k]) / (1 - beta[i,k]);

var du {i in AGENTS, k in COMMODITIES} = # Marginal prices

alpha[i,k] / (1 + x[i,k])^beta[i,k];

Model: cgempec.mod

maximize

objective: sum {i in LEADER} u[i];

subject to

leader_budget {i in LEADER}:

sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

optimality {i in FOLLOWERS, k in COMMODITIES}:

0 <= x[i,k] complements -du[i,k] + p[k] * l[i] >= 0;

budget {i in FOLLOWERS}:

0 <= l[i] complements sum {k in COMMODITIES} p[k]*(e[i,k] - x[i,k]) >= 0;

market {k in COMMODITIES}:

0 <= p[k] complements sum {i in AGENTS} (e[i,k] - x[i,k]) >= 0;

Data: cgempec.dat

set LEADER := Jorge;

set FOLLOWERS := Sven, Todd;

set COMMODITIES := Books, Cars, Food, Pens;

param alpha : Books Cars Food Pens :=

Jorge 1 1 1 1

Sven 1 2 3 4

Todd 2 1 1 5;

param beta (tr): Jorge Sven Todd :=

Books 1.5 2 0.6

Cars 1.6 3 0.7

Food 1.7 2 2.0

Pens 1.8 2 2.5;

Commands: cgempec.cmd

# Load model and data

model cgempec.mod;

data cgempec.dat;

# Specify solver and options

option presolve 0;

option solver "loqo";

# Solve the instance

drop market[’Books’];

fix p[’Books’] := 1;

solve;

# Output results

printf {i in AGENTS, k in COMMODITIES} "%5s %5s: % 5.4e\n", i, k, x[i,k] > cgempec.out;

printf "\n" > cgempec.out;

printf {k in COMMODITIES} "%5s: % 5.4e\n", k, p[k] > cgempec.out;

Output: cgempec.out

Stackleberg Nash Game

Jorge Books: 9.2452e-01 Jorge Books: 8.9825e-01

Jorge Cars: 1.3666e+00 Jorge Cars: 1.4651e+00

Jorge Food: 1.1508e+00 Jorge Food: 1.2021e+00

Jorge Pens: 7.7259e-01 Jorge Pens: 6.8392e-01

Sven Books: 2.5499e-01 Sven Books: 2.5392e-01

Sven Cars: 7.4173e-01 Sven Cars: 7.2054e-01

Sven Food: 1.6657e+00 Sven Food: 1.6271e+00

Sven Pens: 1.4265e+00 Sven Pens: 1.4787e+00

Todd Books: 1.8205e+00 Todd Books: 1.8478e+00

Todd Cars: 8.9169e-01 Todd Cars: 8.1431e-01

Todd Food: 1.8355e-01 Todd Food: 1.7081e-01

Todd Pens: 8.0093e-01 Todd Pens: 8.3738e-01

Books: 1.0000e+00 Books: 1.0000e+00

Cars: 5.9617e-01 Cars: 6.1742e-01

Food: 6.6496e-01 Food: 6.8345e-01

Pens: 1.0700e+00 Pens: 1.0237e+00

Unbounded Multipliers

var z{1..2} >= 0;
var z3;

minimize objf: z[1] + z[2] - z3;
subject to

lin1: -4 * z[1] + z3 <= 0;
lin2: -4 * z[2] + z3 <= 0;
compl: z[1]*z[2] <= 0;

LOQO Output

LOQO 6.06: outlev=2
| Primal | Dual

Iter | Obj Value Infeas | Obj Value Infeas
- - - - - - - - - - - - - - - - - - - - - - - - - -

1 1.000000e+00 0.0e+00 0.000000e+00 1.1e+00
2 6.902180e-01 2.2e-01 -2.672676e-01 2.6e-01
3 2.773222e-01 1.6e-01 -3.051049e-01 1.1e-01

292 -8.213292e-05 1.7e-09 -4.106638e-05 9.1e-07
293 -8.202525e-05 1.7e-09 -4.101255e-05 9.1e-07

294 nan nan nan nan
500 nan nan nan nan

FILTER Output

iter | rho | ||d|| | f / hJ | ||c||/hJt

------+------------+------------+--------------+--------------

0:0 10.0000 0.00000 1.0000000 0.0000000

1:1 10.0000 1.00000 0.0000000 0.0000000

24:1 0.156250 0.196695E-05-0.98347664E-06 0.24180657E-12

25:1 0.156250 0.983477E-06-0.49173832E-06 0.60451644E-13

Norm of KKT residual................ 0.471404521

max( |lam_i| * || a_i ||)........... 2.06155281

Largest modulus multiplier.......... 2711469.25

Limitations

• Multipliers may not exist

• Solvers can have a hard time computing solutions
• Try different algorithms
• Compute feasible starting point

• Stationary points may have descent directions
• Checking for descent is an exponential problem
• Strong stationary points found in certain cases

• Many stationary points – global optimization

• Formulation of follower problem
• Multiple solutions to Nash game
• Nonconvex objective or constraints
• Existence of multipliers

